

1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20
21
22
23
24
25
26
27
28
29
30
31
32
33
34
35
36
37
38
39
40

JSAG (2018), 101–101

dx.doi.org/10.2140/jsag.2018..101

The Journal of Software for
Algebra and Geometry

The ReesAlgebra package in Macaulay2

amended title to match
package name

DAVID EISENBUD

ABSTRACT: This note introduces Rees algebras and some of their uses, with illustrations from version 2.2 of the [Macaulay2](#) package `ReesAlgebra.m2`.

INTRODUCTION. A central construction in modern commutative algebra starts from an ideal I in a commutative ring R , and produces the *Rees algebra*

$$\mathcal{R}(I) := R \oplus I \oplus I^2 \oplus I^3 \oplus \cdots \cong R[It] \subset R[t],$$

where $R[t]$ denotes the polynomial algebra in one variable t over R . For basics on Rees algebras, see [\[Vasconcelos 1994\]](#) and [\[Huneke and Swanson 2006\]](#).

From the point of view of algebraic geometry, the Rees algebra $\mathcal{R}(I)$ is a homogeneous coordinate ring for the graph of a rational map whose total space is the blowup of $\text{Spec } R$ along the scheme defined by I . (In fact, the “Rees algebra” is sometimes called the “blowup algebra”.)

Rees algebras were first studied in the algebraic context by David Rees, in the now-famous paper [\[Rees 1958\]](#). Actually, Rees mainly studied the ring $R[It, t^{-1}]$, now also called the *extended Rees algebra* of I .

Mike Stillman and I, if memory serves, wrote a Rees algebra script for Macaulay classic. It was augmented, and made into the package `ReesAlgebra.m2` around 2002, to study a generalization of Rees algebras to modules described in [\[Eisenbud et al. 2003\]](#). Subsequently Amelia Taylor, Sorin Popescu, the present author, and, at the Macaulay2 Workgroup in July 2017, Ilir Dema, Whitney Liske, and Zhangchi Chen contributed routines for computing many of the invariants of an ideal or module defined in terms of Rees algebras. These routines comprise the package’s primary utility, since Rees algebras of modules other than ideals are comparatively little studied.

We first describe the construction and an example from [\[Eisenbud et al. 2003\]](#). Then we list some of the functionality the package now has and illustrate it with a theorem of Morey and Ulrich. Finally we give examples of how Rees algebras

The author is grateful to the National Science Foundation for partial support.

MSC2010: primary 13A30, 13B22, 13D02; secondary 14C17, 14E15.

Keywords: Rees algebras, blowup, distinguished subvariety, special fiber.

ReesAlgebra.m2 version 2.2

removed duplicated “by”

your memory appears to be correct, according to Mdocs.tar’s docs/man/ap3.tex on <http://www.math.columbia.edu/bayer/> – should I remove this clause?

¹/₂ appear in the Fulton–MacPherson intersection theory and in the resolution of singularities.

³
⁴ **1. THE REES ALGEBRA OF A MODULE.** There are several possible ways of extending the Rees algebra construction from ideals to modules. For simplicity we
⁵ will henceforward only consider finitely generated modules over Noetherian rings.
⁶ Huneke and Ulrich and I argued in [Eisenbud et al. 2003] that the most natural way
⁷ to extend the definition is to think of $R[It]$ as the image of the map of symmetric
⁸ algebras $\text{Sym}(\phi) : \text{Sym}_R(I) \rightarrow \text{Sym}_R(R) = R[t]$, and to generalize it to the case
⁹ of an arbitrary finitely generated module M by setting

$$\mathcal{R}(M) = \text{image } \text{Sym}(\phi),$$

¹² where ϕ is a *versal* map from M to a free module. Such a versal map may be
¹³ computed as the composition of the diagonal embedding

$$M \rightarrow \bigoplus_{i=1}^m M,$$

¹⁶ with the map

$$\bigoplus_{i=1}^m \phi_i : \bigoplus_{i=1}^m M \rightarrow R^m,$$

¹⁹ where ϕ_1, \dots, ϕ_m generate $\text{Hom}_R(M, R)$.

²⁰ Though this is not immediate, the Rees algebra of an ideal in a Noetherian ring,
²¹ in this sense, is the same as the Rees algebra in the classical sense, and in most cases
²² one can take any embedding of the module into a free module in the definition:

²³ **Theorem 1.1** [Eisenbud et al. 2003, Theorems 0.2 and 1.4]. *Let R be a Noetherian*
²⁴ *ring and let M be a finitely generated R -module. Let $\phi : M \rightarrow G$ be a versal map*
²⁵ *of M to a free module. Suppose that ϕ is an inclusion, and let $\psi : M \rightarrow G'$ be any*
²⁶ *inclusion of M into a free module G' . If R is torsion-free over \mathbb{Z} or R is unmixed*
²⁷ *and generically Gorenstein or M is free locally at each associated prime of R ,*
²⁸ *or $G' = R$, then the image of $\text{Sym}(\phi)$ and the image of $\text{Sym}(\psi)$ are naturally*
²⁹ *isomorphic.*

added link, expanded “Thms”

³¹ Nevertheless some examples do violate the conclusion of [Theorem 1.1](#). Here is
³² one from [Eisenbud et al. 2003] in characteristic 5 (any finite characteristic would
³³ work similarly).

³⁴ i1 : p = 5;
³⁵ i2 : R = ZZ/p[x,y,z]/(ideal(x^p,y^p)+(ideal(x,y,z))^(p+1));
³⁶ i3 : M = module ideal(z);

³⁷ It is easy to check that $M \cong R^1/(x, y, z)^p$. We write $\iota : M \rightarrow R^1$ for the embedding
³⁸ as an ideal and ψ for the embedding $M \rightarrow R^2$ sending z to the vector (x, y) .

³⁹ i4 : iota = map(R^1,M,matrix{{z}});
⁴⁰ i5 : psi = map(R^2,M,matrix{{x},{y}});

1 Finally, we choose a versal embedding $M \rightarrow R^3$. It sends z to the vector (x, y, z) :
^{1¹/₂} 2 i6 : phi = versalEmbedding(M);
3 We now compute the kernels of the three maps on symmetric algebras:
4 i7 : Iiota = symmetricKernel iota;
5 i8 : Ipsi = symmetricKernel psi;
6 i9 : Iphi = symmetricKernel phi;
7 and check that the ones corresponding to ϕ and ι are equal, whereas the ones
8 corresponding to ψ and ϕ are not — they differ in degree p .
9 i10 : Iiota == Iphi
10 o10 = true
11 i11 : Ipsi == Iphi
12 o11 = false
13 i12 : numcols basis(p, Iphi)
14 o12 = 3
15 i13 : numcols basis(p, Ipsi)
16 o13 = 1

17 **2. THE REES ALGEBRA AND ITS RELATIONS.** The central routine, `reesIdeal`
18 (with synonym: `reesAlgebraIdeal`), computes an ideal defining the Rees algebra
19 $\mathcal{R}(M)$ as a quotient of a polynomial ring over R from a free presentation of M .
^{20¹/₂} 20 From the Rees ideal we immediately get `reesAlgebra M`. In the case when M
21 is an ideal in R we also compute the important associatedGradedRing $M =$
22 $\mathcal{R}(M)/M$ (and the more geometric sounding but identical `normalCone M`). If I
23 is a (homogeneous) ideal primary to the maximal ideal of a standard graded ring R
24 we compute the Hilbert–Samuel multiplicity of I with the routine `multiplicity`.

multiplicity I → the routine
multiplicity

25 We now describe the basic computation. Suppose that M has a set of generators
26 represented by a map from a free module, and a free presentation

added “and”
is this okay?

$$F \xrightarrow{\alpha} M \rightarrow 0,$$

28 and suppose $F = R^n$. The symmetric algebra of F over R is then a polynomial
29 ring $\text{Sym}_R(F) = R[t_1, \dots, t_n]$ on n new indeterminates t_1, \dots, t_n . By the universal
30 property of the symmetric algebra there is a canonical surjection $\text{Sym}_R(F) \rightarrow$
31 $\text{Sym}_R(M)$, so we may compute the Rees algebra of M as a quotient of $\text{Sym}_R(F)$.
32 The expression removed “the ”

$$I = \text{reesIdeal } M$$

34
35 first uses `versalEmbedding M` to compute a versal map from M to a free module
36 $\beta : M \rightarrow G$. The expression `symmetricKernel $\alpha \circ \beta$` then constructs the map of
37 symmetric algebras $\beta \circ \alpha : \text{Sym}_R(F) \rightarrow \text{Sym}_R(G)$ and uses the built-in Macaulay2
38 routine to compute the kernel

^{39¹/₂} 39
40 $I = \text{reesIdeal } M = \ker \text{Sym}(\beta \circ \alpha) : \text{Sym}_R(F) \rightarrow \text{Sym}_R(G).$

1 There is a different way of computing the Rees algebra that is often much more
 1^{1/2} 2 efficient. It begins by constructing the symmetric algebra of M , and uses the obser-
 3 vation that the construction of the Rees algebra commutes with localization. See
 4 [Eisenbud 1995, Appendix 2] for the necessary facts about symmetric algebras.

5 Suppose that M has a free presentation,

$$6 \quad G \xrightarrow{\alpha} F \xrightarrow{\epsilon} M \rightarrow 0.$$

7 The right exactness of the symmetric algebra functor implies that the symmetric
 8 algebra of M is the quotient of $\text{Sym}_R(F)$ by an ideal I_0 that is generated by the
 9 entries of the matrix

$$10 \quad (t_1 \cdots t_n) \circ \phi,$$

11 (where we have identified ϕ with $\text{Sym}_R(F) \otimes_R \phi$). Thus I_0 is generated by poly-
 12 nomials that are linear in the variables t_i (and because M is the degree 1 part of
 13 $\mathcal{R}(M)$, these are the only linear forms in the t_i in the Rees ideal).

14 If $f \in R$ is an element such that $M[f^{-1}]$ is free on generators g_1, \dots, g_n , it
 15 follows that after inverting f , the Rees algebra of M becomes a polynomial ring
 16 over $R[f^{-1}]$ on indeterminates corresponding to the g_i :

$$17 \quad \mathcal{R}(M)[f^{-1}] = \text{Sym}_R(M[f^{-1}]) = R[G_1, \dots, G_n].$$

18 Now suppose in addition that f is a non-zerodivisor in R . In the diagram
 19

$$20 \quad \begin{array}{ccccc} \text{Sym}_R(F) & \xrightarrow{\alpha} & \text{Sym}_R(M) & \xrightarrow{\beta} & \text{Sym}_R(G) \\ \downarrow & & \downarrow & & \downarrow \\ \text{Sym}_R(F)[f^{-1}] & \xrightarrow{\alpha} & \text{Sym}_R(M)[f^{-1}] & \xrightarrow{\beta} & \text{Sym}_R(G)[f^{-1}] \end{array}$$

21 the two outer vertical maps are inclusions, and it follows that the Rees ideal, which
 22 is the kernel of the map $\mathcal{R}(F) = \text{Sym}_R(F) \rightarrow \mathcal{R}(M)$, is equal to the intersection
 23 of $\mathcal{R}(F)$ with the kernel of

$$24 \quad \text{Sym}_R(F)[f^{-1}] \xrightarrow{\beta} \text{Sym}_R(G)[f^{-1}].$$

25 This intersection may be computed as $I_0 : f^\infty$. The command

$$26 \quad \text{reesIdeal}(I, f)$$

27 computes the Rees ideal in this way.

28 More generally, we say that a module N is *of linear type* if the Rees ideal of M
 29 is equal to the ideal of the symmetric algebra of M ; for example, any complete
 30 intersection ideal is of linear type, and the condition can be tested by the command

$$31 \quad \text{isLinearType } M.$$

32 The procedure above really requires only that f be a non-zerodivisor in R and that
 33 $M[f^{-1}]$ be of linear type over $R[f^{-1}]$.

1 **3. REDUCTIONS AND THE SPECIAL FIBER.** A *reduction* J of an ideal I is a
2 subideal $J \subset I$ over which I is *integrally dependent*. In concrete terms this means
3 that there is some integer r such that $J I^r = I^{r+1}$, and the minimal r with this
4 property is called the reduction number. The property of being a reduction is tested
5 by `isReduction I`, and `reductionNumber I` computes the reduction number. it→the reduction number
6 Now suppose that \mathfrak{m} is a maximal ideal containing I . The special fiber ring is
7 by definition $\mathcal{R}(I)/\mathfrak{m}\mathcal{R}(I)$. It is a standard graded algebra over the field $k := R/\mathfrak{m}$,
8 a quotient of $\text{Sym}_R(F)/\mathfrak{m} = k[t_1, \dots, t_n]$ where, as before, F is a free module of
9 rank n with a surjection to M . The defining ideal of the special fiber ring, and the
10 ring itself, are computed using `specialFiberIdeal I` and `specialFiberRing I`. removed “by” to avoid overfull
line
11 The dimension of the special fiber ring is called the analytic spread of I , usually
12 denoted
13
$$\ell(I) = \text{analyticSpread } I.$$

14 Northcott and Rees [1954] proved that if k is infinite then there always exist re-
15 ductions generated by $\ell(I)$ elements, and this is the minimum possible number;
16 these are called minimal reductions. The smallest possible reduction number for I
17 with respect to a minimal reduction is by definition `reductionNumber I`. (This is reductionnumber →
reductionNumber
18 always achieved by any ideal generated by $\ell(I)$ sufficiently general scalar linear
19 combinations of the generators of I ; but note that when I is homogeneous but has
20 generators of different degrees such linear combinations are sometimes necessarily
21 inhomogeneous.)
22 An interesting special case occurs when R is a graded ring over $k = R_0$ and the
23 generators g_1, \dots, g_n of I are all homogeneous of the same degree. In this case
24 the special fiber ring is easily seen to be equal to the subring $k[g_1, \dots, g_n]$ (usually
25 *not* a polynomial ring) generated by the elements g_i .
26 For example, if I is the ideal of $p \times p$ minors of a $p \times (p + q)$ matrix, then the
27 special fiber ring is equal to the homogeneous coordinate ring \mathbb{G} of the Grassman-
28 nian of p -planes in $p + q$ space. It follows that $\ell(I) = \dim \mathbb{G} = pq + 1$, and the
29 reduction number of I is $(p - 1)(q - 1)$.
30
31 **4. FINDING ELEMENTS OF THE REES IDEAL.** Let M be an R -module and let $\phi :$
32 $R^s \rightarrow R^m$ be its presentation matrix. We identify $\text{Sym}_R(R^m)$ with the polynomial
33 ring $R[t_1, \dots, t_m]$. By the universality of the symmetric algebra construction, the
34 symmetric algebra of I has the form
35
$$\text{Sym}_R(I) = R[t_1, \dots, t_m]/(T\phi),$$

36
37 where we have written T for the vector $(t_1 \dots t_m) \in R[t_1, \dots, t_m]^m$, whose entries
38 correspond to the generators of I , and written $(T\phi)$ for the ideal generated by the
39 entries of the product
40
$$(t_1 \dots t_m)\phi.$$

¹/₂ If $J := (x_1, \dots, x_n) \subset R$ is an ideal containing I , and we write

$$X = (x_1 \cdots x_n) \in R[t_1, \dots, t_m]^n,$$

then there is a matrix ψ defined over $R[t_1, \dots, t_m]$, called the Jacobian dual of ϕ with respect to X , such that $T\phi = X\psi$. (The matrix ψ is generally not unique; Macaulay2 computes it using Gröbner division with remainder.)

If I, J each contain a non-zerodivisor then J will have grade ≥ 1 on the Rees algebra $\mathcal{R}(I)$. Since $(T\phi)$ is contained in the defining ideal of the Rees algebra, the vector X is annihilated by the matrix ψ when regarded over the Rees algebra, and the relation $X\psi \equiv 0$ in $\mathcal{R}(I)$ implies that the $m \times m$ minors of ψ are in the Rees ideal of I .

In very favorable circumstances, one may even have the equality

$$\text{reesIdeal } I == \text{ideal}(T\phi) + \text{minors}(m, \psi).$$

changed math from texttt font, as seems consistent with previous equations; please check

We illustrate with a theorem of Morey and Ulrich. Recall that an ideal I is said to satisfy the condition G_ℓ if the number of generators of the localized ideal I_P is $\leq \text{codim } P$ for every prime ideal P of codimension $< \ell$; equivalently, if I has presentation matrix ϕ as above,

$$\text{codim } I_{m-p}(\phi) > p$$

for $1 \leq p < \ell$.

²⁰/₂ **Theorem 4.1** [Morey and Ulrich 1996]. *Let R be a local Gorenstein ring with infinite residue field, let I be a perfect ideal of grade 2 with m generators, let ϕ be the presentation matrix of I , and let ψ be the Jacobian dual matrix. Let $\ell = \ell(I)$ be the analytic spread. Suppose that I satisfies the condition G_ℓ . The following conditions are equivalent:*

removed “and”

- (1) $\mathcal{R}(I)$ is Cohen–Macaulay and $I_{(m-\ell)}(\phi) = I_1(\phi)^{m-\ell}$.
- (2) $r(I) < \ell$ and $I_{m+1-\ell}\phi = (I_1\phi)^{m+1-\ell}$.
- (3) The ideal of $\mathcal{R}(I)$ is equal to the sum of the ideal of $\text{Sym}(I)$ with the Jacobian dual minors, $I_m\psi$.

We can check all these conditions with functions in the package. We start with the presentation matrix ϕ of an $m=n+1$ -generator perfect ideal such that the first row consists of the n variables of the ring, and the rest of the rows are reasonably general (in this case random quadrics):

whose→the

```

i2 : setRandomSeed 0
i3 : n=3;
i4 : kk = ZZ/101;
i5 : S = kk[a_0..a_(n-2)];
i6 : phi = transpose map(S^(n-1), S^{-1, (n-1):-2},
    (i,j) -> if j == 0 then a_i else random(2,S));
3      2

```

```

1  o6 : Matrix S <--- S
11/2 2 i7 : I = minors(n-1,phi);
3  This is a perfect codimension 2 ideal, as we see from the Betti table:
4  i8 : betti (F = res I)
5          0 1 2
6  o8 = total: 1 3 2
7          0: 1 . .
8          1: . . .
9          2: . 2 .
10         3: . 1 2
11  We compute the analytic spread  $\ell$  and the reduction number  $r$ :
12 i12 : ell = analyticSpread I
13 o12 = 2
14 i13 : r = reductionNumber(I, minimalReduction I)
15 o13 = 1
16 Now we can check the condition  $G_\ell$ , first probabilistically:
17 i15 : whichGm I >= ell
18 o15 = true
19 and now deterministically:
20 201/2 i17 : apply(toList(1..ell-1),
21              p-> {p+1, codim minors(n-p, phi)})
22 o17 = {{2, 2}}
23
24 We now check the three equivalent conditions of the Morey–Ulrich theorem. Since
25  $\ell = n - 1$  in this case, the second parts of conditions (1) and (2) are vacuously
26 satisfied, and since  $r < \ell$  the conditions must all be satisfied. We first check that
27  $\mathcal{R}(I)$  is Cohen–Macaulay:
28 i19 : reesI = reesIdeal I;
29 o19 : Ideal of S[w , w , w ]
30          0 1 2
31 i20 : codim reesI
32 o20 = 2
33 i21 : betti res reesI
34          0 1 2
35 o21 = total: 1 3 2
36          0: 1 . .
37          1: . . .
38          2: . 2 .
39          3: . 1 2
391/2 39 Finally, we wish to see that reesIdeal  $I$  is generated by the ideal of the symmetric
40 algebra together with the Jacobian dual:

```

part \rightarrow parts
is \rightarrow are

reesIdeal $I \rightarrow$ reesIdeal I

```

1 i23 : psi = jacobianDual phi;
11/2 2 o23 : Matrix (S[w , w , w ]) <--- (S[w , w , w ])
3      0 1 2      0 1 2
4
5 We now compute the ideal  $J$  of the symmetric algebra; we do this by hand, since rearranged to avoid overfull
6 the command symmetricAlgebra  $I$  would return the ideal over a different ring. line
7 i25 : ST = ring psi
8 i26 : T = vars ST
9 o26 = | w_0 w_1 w_2 |
10 i27 : J = ideal(T*promote(phi, ST))
11 i28 :      betti res J
12      0 1 2
13 o28 = total: 1 2 1
14      0: 1 . .
15      1: . . .
16      2: . 2 .
17      3: . . .
18      4: . . 1
19 i29 : J1 = minors(ell, psi)
20
21 We compute the resolution of  $G := J + J1$ , to see that the resulting ideal is perfect,
22 which also shows that it is the full ideal of the Rees algebra. We also check directly
23 that it has the same resolution as the computed Rees ideal of  $I$ :
24
25 201/2 i30 : betti (G = res trim (J+J1))
26      0 1 2
27 o30 = total: 1 3 2
28      0: 1 . .
29      1: . . .
30      2: . 2 .
31      3: . 1 2
32
33 i31 : betti res reesIdeal I
34      0 1 2
35 o31 = total: 1 3 2
36      0: 1 . .
37      1: . . .
38      2: . 2 .
39      3: . 1 2
40

```

5. DISTINGUISHED SUBVARIETIES. The key construction in the Fulton–MacPherson definition of the refined intersection product [Fulton 1998, Section 6.1] involves normal cones, and is easy to implement using the tools in this package. The simplest case is the intersection of two subvarieties $X, V \subset Y$. If X and V meet in the *expected dimension*, defined to be $\dim V - \operatorname{codim}_Y X$, and the ambient variety Y is smooth, then one can assign multiplicities m_i to the components W_i of $X \cap V$, and the intersection product has the form $[X][V] = \sum m_i[W_i]$. The astonishing result of the Fulton–MacPherson theory is that if $X \subset Y$ is locally a complete

¹/₂ intersection, then, no matter how singular Y and no matter how strange the actual intersection $X \cap V$, the intersection product $X \cdot V$ can be given a meaning as a rational equivalence class of cycles of the expected dimension on X , or even on certain *distinguished* subvarieties Z_i of $X \cap V$. This class comes with a canonical decomposition $\sum_i m_i \alpha_i$, where the m_i are positive integers, and α_i is a cycle of the expected dimension (possibly 0) on $Z_i \subset X \cap V$ (the same Z_i can appear several times, with different multiplicities and cycles).

In the general case, the subvariety V is replaced by a morphism $f : V \rightarrow Y$ from a variety V , and this is the key to the functoriality of the intersection product. The routines in this package work in the general setting, but for simplicity we will stick with the basic case in this description.

We now describe the distinguished subvarieties and their multiplicities. This part of the construction sheafifies, so (as in the package) we work in the affine case. We do not require any hypothesis on X , Y or V .

Let S be a ring (for example, the coordinate ring of Y) and let $I \subset S$ be an ideal (for example, the ideal of X). Write

$$T := \text{gr}_I S = S/I \oplus I/I^2 \oplus \dots$$

for the associated graded ring of I , and let π be the inclusion of S/I into T as the degree 0 part.

²⁰/₂ Let $f : S \rightarrow R$ be a ring homomorphism (for example, representing the projection $S \rightarrow S/(I(V))$). Let $K \subset T$ be the kernel of the induced map $\text{gr}_I S \rightarrow \text{gr}_{f(I)R} R$.

Let P_1, \dots, P_m be the minimal primes over K in $\text{gr}_I R$. We define p_i to be the degree 0 part of P_i ; that is, $p_i := P_i \cap S/I$. These are the distinguished prime ideals of S/I , and they clearly contain the kernel of $\bar{f} : S/I \rightarrow R/f(I)R$, so in the case where $R = S/J$ they contain $I + J$. Thus, in this case, they represent subvarieties of $X \cap V$.

Let m_i be the multiplicity with which P_i appears in the primary decomposition of K — that is,

$$m_i := \text{length}_{t_{P_i}} P_i / K_{P_i}.$$

Returning to the geometric language, and the case where $X \subset Y$ is locally a complete intersection in a quasiprojective variety, the cycle class α_i in the Chow group of the variety Z_i corresponding to p_i is defined as the Gysin image of the class of the subvariety corresponding to P_i in the projectivized normal bundle of X in Y — a construction not included in this package.

Here are some simple examples in which `distinguished` is used to compute the distinguished varieties of intersections in \mathbb{A}^n , via the function `intersectInP`. First, the familiar multiplicity 2 intersection of a conic with a tangent line.

³⁹/₂ i2 : kk = ZZ/101;
i3 : P = kk[x,y];

1 i4 : I = ideal "x2-y"; J=ideal y;
 1^{1/2} 2 i6 : intersectInP(I,J)
 3 o6 = {{2, ideal (y, x)}}

4 Slightly more interesting, the following shows what happens when the intersections
 5 aren't rational:

6 i7 : I = ideal "x4+y3+1";
 7 i8 : intersectInP(I,J)
 8 o8 = {{1, ideal (y, x² + 10)}, {1, ideal (y, x² - 10)}}

10 The real interest in the construction is in the case of improper intersections. Here
 11 are some typical results:

12 i9 : I = ideal "x2y"; J=ideal "xy2";
 13 i11 : intersectInP(I,J)
 14 o11 = {{2, ideal x}, {5, ideal (y, x)}, {2, ideal y}}
 15 i12 : intersectInP(I,I)
 16 o12 = {{1, ideal y}, {4, ideal x}, {4, ideal (y, x)}}

17 **6. REES ALGEBRAS AND DESINGULARIZATION.** We conclude with an example
 18 illustrating a general result about projective birational maps of varieties. Recall
 19 that a map $B \rightarrow X$ of varieties is projective if it is the composition of a closed
 20^{1/2} embedding $B \subset X \times \mathbb{P}^n$ with the projection to X . It is birational if it is generically
 21 an isomorphism. The inclusion of a ring into the Rees algebra of an ideal corre-
 22 sponds to a map from proj of the Rees algebra to Spec of the ring, called a blowup,
 23 that is such a proper birational transformation, and in fact every proper birational
 24 transformation to an affine variety (or more generally to any scheme, if one works
 25 with sheaves of ideals) can be realized in this way.

varieites → varieties

Proj → proj
 spec → Spec
 for consistency with use
 elsewhere in the paper; is this
 correct?

26 The theorem of embedded resolution of singularities (proven by Hironaka in
 27 characteristic 0 and conjectured in general) says that, given any subvariety X of a
 28 smooth variety Y , there is a finite sequence of blowups

$$B_n \rightarrow \cdots \rightarrow B_2 \rightarrow B_1 \rightarrow Y$$

31 of smooth subvarieties that lie over the singular set of X , and a component of the
 32 preimage of X in B_n that is smooth and maps birationally to X . In the case of
 33 plane curves, this can be done with a sequence of blowups of closed points. But
 34 in fact *any* sequence of blowups of a quasiprojective variety can be replaced with
 35 a single blowup [Hartshorne 1977, Theorem II.7.17] of a more complicated ideal.
 36 We illustrate this with the desingularization of a tacnode (the union of two smooth
 37 curves that meet with a simple tangency).

any → any

removed unmatched "("
 added "this"

39^{1/2} **Example 6.1.** Blowing-up (x^2, y) in $k[x, y]$ desingularizes the tacnode $x^2 - y^4$ in
 40 a single step.

```

1  i1 : R = ZZ/32003[x,y];
11/2 2  i2 : tacnode = ideal(x^2-y^4);
3  i3 : mm = ideal(x,y^2);
4  i4 : B = first flattenRing reesAlgebra mm;
5  i5 : irrelB = ideal(w_0,w_1);
6  i6 : proj = map(B,R,{x,y});
7  i7 : totalTransform = proj tacnode
8      4      2
9  o7 = ideal(- y  + x )
10 i8 : netList (D = decompose totalTransform)
11 +-----+
12 | ideal (y, x) |
13 +-----+
14 |      2      |
15 | ideal (y  + x, w  + w ) |
16 |      0      1 |
17 +-----+
18 |      2      |
19 | ideal (y  - x, w  - w ) |
20 |      0      1 |
21 +-----+
22 i9 : exceptional = proj mm
23      2
24 o9 = ideal (x, y )
201/2 25 i10 : strictTransform = saturate(
26      totalTransform, exceptional);
27 i11 : netList decompose strictTransform
28 +-----+
29 |      2      |
30 | ideal (y  + x, w  + w ) |
31 |      0      1 |
32 +-----+
33 |      2      |
34 | ideal (y  - x, w  - w ) |
35 |      0      1 |
36 +-----+
37 i12 : sing0 = sub(ideal singularLocus strictTransform, B);
38 i13 : sing = saturate(sing0,irrelB)
39 o13 = ideal 1

```

33 The last line asserts that the singular locus of the strict transform is empty; that is, removed duplicated “the”s
34 the scheme defined by strictTransform is smooth (in this case it is the union of “strictTransform” →
35 two disjoint smooth curves). strictTransform

36 SUPPLEMENT. The [online supplement](#) contains version 2.2 of ReesAlgebra.m2.

37 REFERENCES.

38 [Eisenbud 1995] D. Eisenbud, *Commutative algebra with a view toward algebraic geometry*, Grad-
39 uate Texts in Mathematics **150**, Springer, 1995. MR Zbl

added this
[EU], [KU], [U] and [VV]
were not referred to and don't
appear here. If you'd like to
reinstate any of them, please
cite it somewhere in the text to
make its relevance clear.

- 1 [Eisenbud et al. 2003] D. Eisenbud, C. Huneke, and B. Ulrich, “What is the Rees algebra of a
2 module?”, *Proc. Amer. Math. Soc.* **131**:3 (2003), 701–708. [MR](#) [Zbl](#)
- 3 [Fulton 1998] W. Fulton, *Intersection theory*, 2nd ed., Ergebnisse der Mathematik (3) **2**, Springer,
4 1998. [MR](#) [Zbl](#)
- 5 [Hartshorne 1977] R. Hartshorne, *Algebraic geometry*, Graduate Texts in Mathematics **52**, Springer,
6 1977. [MR](#) [Zbl](#)
- 7 [Huneke and Swanson 2006] C. Huneke and I. Swanson, *Integral closure of ideals, rings, and mod-* rearranged order of authors
8 *ules*, London Mathematical Society Lecture Note Series **336**, Cambridge University Press, 2006.
9 [MR](#) [Zbl](#)
- 10 [Morey and Ulrich 1996] S. Morey and B. Ulrich, “Rees algebras of ideals with low codimension”,
11 *Proc. Amer. Math. Soc.* **124**:12 (1996), 3653–3661. [MR](#) [Zbl](#)
- 12 [Northcott and Rees 1954] D. G. Northcott and D. Rees, “Reductions of ideals in local rings”, *Proc.*
13 *Cambridge Philos. Soc.* **50** (1954), 145–158. [MR](#) [Zbl](#)
- 14 [Rees 1958] D. Rees, “On a problem of Zariski”, *Illinois J. Math.* **2** (1958), 145–149. [MR](#) [Zbl](#)
- 15 [Vasconcelos 1994] W. V. Vasconcelos, *Arithmetic of blowup algebras*, London Mathematical Soci-
16 ety Lecture Note Series **195**, Cambridge University Press, 1994. [MR](#) [Zbl](#)

RECEIVED: 19 Aug 2017 REVISED: 4 Feb 2018 ACCEPTED: 21 May 2018

DAVID EISENBUD:

de@msri.org

Mathematical Sciences Research Institute, Berkeley, CA, United States